Math 277
Ordinary Differential Equations

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 6

Exact Equations

Initializations

```
> restart;
```

6.1 Exact Equations.

Exact differential equations are of the form

$$\frac{\partial F}{\partial x}\, dx + \frac{\partial F}{\partial y}\, dy = 0$$

In class we will show that the equation

$$M(x, y)\, dx + N(x, y)\, dy = 0$$

is exact on a rectangle R, if M and N have continuous first order partial derivatives on R and

$$\frac{\partial M}{\partial y} = \frac{\partial N}{\partial x}$$

for all $(x, y) \in R$.

Examples

Example 5.1.1

Show that the equation

$$\left(y e^{xy} \cos 2x - 2 e^{xy} \sin 2x + 2x\right) dx + \left(x e^{xy} \cos 2x - 3\right) dy = 0$$

is exact. Then solve the equation.

Solution

We apply the test for exactness.

```
> M:=y*exp(x*y)*cos(2*x)-2*exp(x*y)*sin(2*x)+2*x;
  N:=x*exp(x*y)*cos(2*x)-3;
```

$$M := y\, e^{xy} \cos(2x) - 2\, e^{xy} \sin(2x) + 2x$$

$$N := x\, e^{xy} \cos(2x) - 3 \qquad \textbf{(2.1.1.1)}$$

Clearly, both M and N have continuous first order partial derivatives on the entire xy plane.
We now verify that

$$\frac{\partial M}{\partial y} = \frac{\partial N}{\partial x}$$

```
> e1:=Diff(M,y)-Diff(N,x);
  test:=simplify(value(e1));
```

$$e1 := \frac{\partial}{\partial y}\left(y\,e^{xy}\cos(2\,x) - 2\,e^{xy}\sin(2\,x) + 2\,x\right) - \left(\frac{\partial}{\partial x}\left(x\,e^{xy}\cos(2\,x) - 3\right)\right)$$

$$test := 0 \qquad (2.1.1.2)$$

We conclude that the equation is exact.

To obtain the solution to this exact equation we integrate M with respect to the variable x.

```
> e1:=Int(M,x)+g(y);
  F:=simplify(value(e1));
```

$$e1 := \int \left(y\,e^{xy}\cos(2\,x) - 2\,e^{xy}\sin(2\,x) + 2\,x\right) dx + g(y)$$

$$F := e^{xy}\cos(2\,x) + x^2 + g(y) \qquad (2.1.1.3)$$

Set $\frac{\partial F}{\partial y}$ equal to N and solve for $\frac{dg}{dy}$.

```
> e2:=Diff(F, y)=N;
  e3:=value(e2);
  e4:=isolate(e3, diff(g(y),y));
```

$$e2 := \frac{\partial}{\partial y}\left(e^{xy}\cos(2\,x) + x^2 + g(y)\right) = x\,e^{xy}\cos(2\,x) - 3$$

$$e3 := x\,e^{xy}\cos(2\,x) + \frac{d}{dy}\,g(y) = x\,e^{xy}\cos(2\,x) - 3$$

$$e4 := \frac{d}{dy}\,g(y) = -3 \qquad (2.1.1.4)$$

Finally, we compute $g(y)$ and update F.

```
> e5:=map(int, e4, y);
```

$$e5 := g(y) = -3\,y \qquad (2.1.1.5)$$

```
> F:=eval(F, e5);
```

$$F := e^{xy}\cos(2\,x) + x^2 - 3\,y \qquad (2.1.1.6)$$

The implicit solutions are given by $F(x, y) = C$.

```
> sol:=F=C;
```

$$sol := e^{xy}\cos(2\,x) + x^2 - 3\,y = C \qquad (2.1.1.7)$$